

	
3GPP TSG-RAN5 Meeting #2021-TTCN email	R5s211472
Online, , 7th Dec 2020 - 31st Dec 2021
	CR-Form-v12.1

	CHANGE REQUEST

	

	
	38.523-3
	CR
	2111
	rev
	-
	Current version:
	17.0.0
	

	

	For HELP on using this form: comprehensive instructions can be found at 
http://www.3gpp.org/Change-Requests.

	



	Proposed change affects:
	UICC apps
	
	ME
	
	Radio Access Network
	
	Core Network
	



	

	Title:	
	Addition of NR5GC test case 8.1.5.8.1 in FR1

	
	

	Source to WG:
	Datang LinkTester Technology

	Source to TSG:
	R5

	
	

	Work item code:
	5GS_NR_LTE-UEConTest
	
	Date:
	2021-10-09

	
	
	
	
	

	Category:
	B
	
	Release:
	Rel-17

	
	Use one of the following categories:
F  (correction)
A  (mirror corresponding to a change in an earlier 													release)
B  (addition of feature), 
C  (functional modification of feature)
D  (editorial modification)
Detailed explanations of the above categories can
be found in 3GPP TR 21.900.
	Use one of the following releases:
Rel-8	(Release 8)
Rel-9	(Release 9)
Rel-10	(Release 10)
Rel-11	(Release 11)
…
Rel-15	(Release 15)
Rel-16	(Release 16)
Rel-17	(Release 17)
Rel-18	(Release 18)

	
	

	Reason for change:
	To add Rel-15 NR5GC test case 8.1.5.8.1 to the NR5GC ATS

	
	

	Summary of change:
	This document lists all changes applied to NR5GC test case  8.1.5.8.1　required for approval. See detailed change description for further information.

	
	

	Consequences if not approved:
	Test case will not be added to ATS.

	
	

	Clauses affected:
	8.1.5.8.1.NR5GC

	
	

	
	Y
	N
	
	

	Other specs
	
	X
	 Other core specifications	
	TS/TR ... CR ... 

	affected:
	X
	
	 Test specifications
	38.523-1

	(show related CRs)
	
	X
	 O&M Specifications
	TS/TR ... CR ... 

	
	

	Other comments:
	

	
	

	This CR's revision history:
	



Page 1


[bookmark: _Toc84670982]Table of Contents
Table of Contents	2
1	Overview	3
1.1	Verification Test Summary	3
2	Corrections required	3
2.1	Change 1	3
2.2	Change 2	12
3	Branches executed	18
4	Execution Log Files	18
4.1	Media Tek M80	18
5	References	18


1. [bookmark: _Toc122434485][bookmark: _Toc84670983]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 8.1.5.8.1 which is part of the NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Pengchong Shen
	shenpengchong@datangmobile.cn	                             
0. [bookmark: _Toc84670984]Verification Test Summary
Test Case: 	8.1.5.8.1
ATS Version:	iwd-TTCN3-B2020-09_D21wk37
System Simulator used:	Datang LinkTester 5G Protocol Conformance Test System ECT9610
UE used:	Media Tek M80
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc84670985]Corrections required
1. [bookmark: _Toc84670986][bookmark: OLE_LINK4][bookmark: OLE_LINK5][bookmark: OLE_LINK6]Change 1
	Function name
	f_TC_8_1_5_8_1_NR5GC_TestBody

	Reason for change
	1. If  the UE capability information is too big, it will be segmented. So it shall need more 1 subframe to receive UE capability information.
2. At step11-12, according to 38.508 Table 4.6.1-2 count-Uplink and  count-Downlink should be not checked.
3. For message of deregistration in ULInformationTransfer  is transformed  by SRB2, but the current TTCN dose not resume SRB2, which make UE dose not transform deregistrastion message. 
4. It miss the procedure of service accept at step 9 – 10, which will fail the test case when cheking the status of security during turn off.

Note: it will be a prose CR to support.


	Summary of change
	1. Add a new input parameter p_If_UEinfo for function f_NR5GC_RRC_ProcedureDelay,  and if the message checked is UE capability inforrmation, add 20 slots for v_Nslots.
2. Add new variable v_CountList_UE for cr_38508_CounterCheckResponse.
3. Add f_NR5GC_RRCReestablishment_Steps8To13 after Step 25 – 26.
4. Add service sccept in rrc reconfiguration at step9 – 10.

	TTCN module
	RRC_Others_NR5GC.ttcn

	MCC160 Comment
	


Before Change
	  function f_TC_8_1_5_8_1_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var RNTI_Value v_RNTI_Value;
    var PhysCellId v_PhysCellId;
    var ShortMAC_I v_ShortMAC_I;
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters;
    var NR_SecurityParams_Type v_SecurityParams;
    var octetstring v_EncodeCapRequestNR;
    var template (value) DRB_CountMSB_InfoList v_CountList;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) CellGroupConfig v_CellGroupConfig;
    var NR_RadioBearerList_Type v_SS_Config;
    var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;
    var template (value) NR_RRC_MSG_Request_Type v_RrcMsg;
    var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
    var template(omit) octetstring v_EncodedSCG_Config := omit;
    var NR_DRBInfoList_Type v_ExistingDRBs;
    var template(value) NR_RachProcedureConfig_Type v_NR_RachProcedureConfig;

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;


    v_CellPowerList_AtT0 :={
      cs_NR_CellPower(nr_Cell2, -96, -96) //FFS in prose for FR2
    };

    v_CellPowerList_AtT1:={
      cs_NR_CellPower(nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),   //FFS in prose for FR2
      cs_NR_CellPower(nr_Cell2, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)          //FFS in prose for FR2
    };

    f_NR_SetCellPowerList(v_CellPowerList_AtT0);

    //@siclog "Step 1" siclog@
    //The SS transmits a Paging message including a matched ng-5G-S-TMSI.
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE);

    //@siclog "Step 2" siclog@
    //The UE transmits an RRCSetupRequest message.
    f_NR_RRC_SetupRequest_Def(nr_Cell1);

    //@siclog "Steps 3 - 4" siclog@
    //The SS transmits an RRCSetup message.
    //The SS starts transmitting a periodic DCI from (10*2microsec+1+TDL) slots after step 3 to schedule PUSCH.
    //Check: Does the UE transmit an RRCSetupComplete message within (10*2microsec+1+TDL+K2) slots15 (FDD)/18 (TDD) subframes after successful completion of step 3?
    v_ReceivedNR_SRB_COMMON_IND := f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                                                              cas_NR_SRB0_RrcPdu_REQ(nr_Cell1,cs_TimingInfo_Now,f_NR_38508_RRCSetup(nr_Cell1, tsc_NR_RRC_TI_Def)),
                                                              car_NR_SRB1_RrcNasPdu_IND(nr_Cell1, cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *), cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected)),
                                                              10, "Step 4" );

    //@siclog "Step 5 - 6" siclog@
    //The SS transmits a SecurityModeCommand message.
    //The SS starts transmitting a periodic DCI from (5*2microsec+1+TDL) slots after step 5 to schedule PUSCH.
    //Check: Does the UE transmit a SecurityModeComplete message within (5*2microsec+1+TDL+K2) 10 (FDD)/13 (TDD) subframes slots after successful completion of step 5?
    v_SecurityParams := f_NR_Security_Get();
    v_SecurityParams := f_NR_Authentication_InitAS (v_SecurityParams, v_ReceivedNR_SRB_COMMON_IND.Signalling.Nas[0].SecurityProtection.NasCount);
    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(nr_Cell1);
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(nr_Cell1,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);

    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,
                                                      -,
                                                      cs_38508_RRCSecurityModeCommand(tsc_NR_RRC_TI_Def,
                                                                                      cs_38508_SecurityAlgorithmConfig(v_SecurityParams.AS_Ciphering.Algorithm,
                                                                                                                       v_SecurityParams.AS_Integrity.Algorithm))),
                               car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCSecurityModeComplete(tsc_NR_RRC_TI_Def)),
                               5, "Step 6" );

    f_NR_Security_Set(v_SecurityParams);

    //@siclog "Step 7 - 8" siclog@
    //The SS transmits a UECapabilityEnquiry message.
    //The SS starts transmitting a periodic DCI from (80*2microsec+1+TDL) slots after step 7 to schedule PUSCH.
    //Check: Does the UE transmit a UECapabilityInformation message within (80*2microsec+1+TDL+K2) slots after successful completion of step 7?
    v_EncodeCapRequestNR := bit2oct(encvalue( cs_38508_UECapabilityRequestFilterNR({ cs_NR_BandInformationNR(f_NR_CellInfo_GetBandIndicator(nr_Cell1)) }) ));
    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_UECapabilityEnquiry(tsc_NR_RRC_TI_Def, nr, v_EncodeCapRequestNR)),
                               car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_UECapabilityInformationAny(tsc_NR_RRC_TI_Def)),
                               80, "Step 8" );

    //@siclog "Step 9 - 10" siclog@
    //The SS transmits an RRCReconfiguration message to establish SRB2 and DRB #n for the UE
    fl_NR_BuildRRCReconfigExistingDRBs (v_RadioBearerConfig, v_CellGroupConfig, v_SS_Config, nr_Cell1);
    
    v_SecurityParams := f_NR_Security_Get();
    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(v_CellGroupConfig)));

    //Reconfigure DRB in SS - existing DRBs are not reconfigured during release
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, v_SS_Config, cs_TimingInfo_Now);
    
    // Configure PDCP security
    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(nr_Cell1, false);
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(nr_Cell1,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);


    v_RrcMsg := cs_NR_RRC_MSG_Request_DCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig, -, -, v_V1530Ext));
    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_RrcNasPduList_REQ(nr_Cell1, tsc_NR_RbId_SRB1, cs_TimingInfo_Now, v_RrcMsg),
                               car_NR_SRB_RrcPdu_IND(nr_Cell1, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete),
                               10, "Step 10" );

    //@siclog "Step 11 - 12" siclog@
    //The SS transmits a CounterCheck message.
    //The SS starts transmitting a periodic DCI from (5*2microsec+1+TDL) slots after step 11 to schedule PUSCH.
    //Check: Does the UE transmit a CounterCheckResponse message within (5*2microsec+1+TDL+K2)10 (FDD)/13 (TDD) subframes slots after successful completion of step 11?
    v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();
    v_CountList := {cs_38508_DRB_CountMSB_Info(v_ExistingDRBs[0].DRB_Id, 1, 1)};
    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_CounterCheck(tsc_NR_RRC_TI_Def, v_CountList)),
                               car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_CounterCheckResponse(tsc_NR_RRC_TI_Def, v_CountList)),
                               5, "Step 12" );

    //@siclog "Step 13" siclog@
    //Void

    //@siclog "Step 14a1-14a7" siclog@
    if ( pc_inactiveState ){

    //@siclog "Step 14a1" siclog@
    //The SS transmits an RRCRelease message with suspendConfig.
    f_NR_RRC_InactiveDef(nr_Cell1);

    //@siclog "Step 14a2" siclog@
    //Wait 1s to let UE enter RRC_INACTIVE state.
    f_Delay(1.0);

    //@siclog "Step 14a3" siclog@
    //The SS transmits a Paging message including a matched ng-5G-S-TMSI.
    f_NR_Paging_Def(nr_Cell1, RRC_INACTIVE);

    v_CellGroupConfig := f_NR_GetSCGConfigResume(nr_Cell1);
    v_EncodedSCG_Config := bit2oct(encvalue(v_CellGroupConfig));

    //@siclog "Step 14a4" siclog@
    //The UE transmits an RRCResumeRequest message
    SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, mt_Access)));

    //@siclog "Step 14a5" siclog@
    //The SS transmits an RRCResume message for the UE

    if(pc_reducedCP_Latency){
        //@siclog "Step 14a5Aa1 - 14a5Aa2" siclog@
        //The SS starts transmitting a periodic DCI from (10* 2 microsec +1+TDL) slots after step 14a5 to schedule PUSCH.
        //Check: Does the UE transmit an RRCResumeComplete message within (6*2microsec+1+TDL+K2) slots after successful completion of step 14a5?
        f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                                   cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, cs_38508_RRCResume(tsc_NR_RRC_TI_Def, v_EncodedSCG_Config, v_RadioBearerConfig)),
                                   car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeComplete),
                                   6, "Step 14a5Aa2" );
    }else
    {
        //@siclog "Step 14a5Ab1 - 14a5Ab2" siclog@
        //The SS starts transmitting a periodic DCI from n slots after step 14a5 to schedule PUSCH.
        //Check: Does the UE transmit an RRCResumeComplete message within (10* 2 microsed +1+TDL+K2)slots after successful completion of step 14a5?
        f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                                   cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, cs_38508_RRCResume(tsc_NR_RRC_TI_Def, v_EncodedSCG_Config, v_RadioBearerConfig)),
                                   car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeComplete),
                                   10, "Step 14a5Ab2" );
    }
    v_NR_RachProcedureConfig := f_NR_InitialiseRachProcedure(nr_Cell1);
    f_NR_SS_RachProcedureConfig(nr_Cell1, v_NR_RachProcedureConfig);

    //@siclog "Step 14a6-14a7" siclog@
    //Void
    }
    //@siclog "Step 14-21" siclog@
    //Void

    //@siclog "Step 22" siclog@
    //The SS re-adjusts the SS levels according to row "T1" in table 8.1.5.8.1.3.2-1/2 in order that the radio link quality of NR Cell 1 is degraded, and NR Cell 2 is suitable for camping
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 23" siclog@
    //Wait 1s to let T310 expire
    f_Delay(1.0);

    //@siclog "Step 24" siclog@
    //The UE transmits an RRCReestablishmentRequest message on Cell 2

    v_RNTI_Value := f_NR_CellInfo_GetRNTI(nr_Cell1);
    v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
    v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(nr_Cell2);
        
    f_NR_CellInfo_SetRNTI(nr_Cell2, tsc_C_RNTI_Value4);
    f_NR_SS_C_RNTI_Config(nr_Cell2, tsc_C_RNTI_Value4, -, tsc_NoCnfReq, tsc_FollowOnFlag);
    f_NR_SS_CommonCellConfig(nr_Cell2, cads_NR_RachProcedure_DcchDtchToCSS_Config_REQ(nr_Cell2, cs_TimingInfo_Now, tsc_CnfReq,
                                                                                      cs_NR_RachProcedureConfig_CBRA_Msg4Based(v_NR_PhysicalParameters, cs_NR_RachProcedureMsg4WithNoRrcMsg),
                                                                                      cs_NR_SearchSpaceDlDciAssignment_CSSType1(v_NR_PhysicalParameters),
                                                                                      cs_NR_SearchSpaceUlDciAssignmentCSS(v_NR_PhysicalParameters, cs_NR_DciFormat_0_0_Params)));

    v_ShortMAC_I := f_NR_Calculate_ShortMAC(nr_Cell1, nr_Cell2);

    SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell2, cr_38508_RRCReestablishmentRequest(cr_NR_ReestabUE_Identity (v_RNTI_Value, v_PhysCellId, v_ShortMAC_I), otherFailure)));
    

    //@siclog "Step 25 - 26 " siclog@
    //The SS transmits an RRCReestablishment message
    //The SS starts transmitting a periodic DCI from (10*2microsec+1+TDL) slots after step 25 to schedule PUSCH.
    //Check: Does the UE transmit an RRCReestablishmentComplete message within (10*2microsec+1+TDL+K2) 15 (FDD)/18 (TDD) subframes slots after successful completion of step 26?
   
    f_NR_ULGrantConfiguration_Start(nr_Cell2, -, -, -, cs_NR_UplinkTimeAlignment_Stop);
    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell2, 0, -, true);

   f_NR5GC_RRC_ProcedureDelay(nr_Cell2,
                              cas_NR_SRB1_RrcPdu_REQ(nr_Cell2, cs_TimingInfo_Now, cs_38508_RRCReestablishment_Def(tsc_NR_RRC_TI_Def, 0)),
                              car_NR_SRB1_RrcPdu_IND ( nr_Cell2, cr_38508_RRCReestablishmentComplete),
                              10, "Step 26" );

    
   //@siclog "Step 27 - 35" siclog@
   //Void

 }


After Change
	function f_TC_8_1_5_8_1_NR5GC_TestBody() runs on NR5GC_PTC
{
  var RNTI_Value v_RNTI_Value;
  var PhysCellId v_PhysCellId;
  var ShortMAC_I v_ShortMAC_I;
  var NR_PhysicalParameters_Type v_NR_PhysicalParameters;
  var NR_SecurityParams_Type v_SecurityParams;
  var octetstring v_EncodeCapRequestNR;
  var template (value) DRB_CountMSB_InfoList v_CountList;
  var template (value) RadioBearerConfig v_RadioBearerConfig;
  var template (value) CellGroupConfig v_CellGroupConfig;
  var NR_RadioBearerList_Type v_SS_Config;
  var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;
  var template (value) NR_RRC_MSG_Request_Type v_RrcMsg;
  var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
  var template(omit) octetstring v_EncodedSCG_Config := omit;
  var NR_DRBInfoList_Type v_ExistingDRBs;
  var template(value) NR_RachProcedureConfig_Type v_NR_RachProcedureConfig;

  var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;
  var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;
  var template (value) NG_NAS_MSG_RequestList_Type v_ServerAcceptMsg;  
  var template DRB_CountInfoList v_CountList_UE;


  v_CellPowerList_AtT0 :={
	cs_NR_CellPower(nr_Cell2, -96, -96) //FFS in prose for FR2
  };

  v_CellPowerList_AtT1:={
	cs_NR_CellPower(nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),   //FFS in prose for FR2
	cs_NR_CellPower(nr_Cell2, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)		   //FFS in prose for FR2
  };

  f_NR_SetCellPowerList(v_CellPowerList_AtT0);

    //@siclog "Step 1" siclog@
  //The SS transmits a Paging message including a matched ng-5G-S-TMSI.
  f_NR_Paging_Def(nr_Cell1, RRC_IDLE);

    //@siclog "Step 2" siclog@
  //The UE transmits an RRCSetupRequest message.
  f_NR_RRC_SetupRequest_Def(nr_Cell1);

    //@siclog "Steps 3 - 4" siclog@
  //The SS transmits an RRCSetup message.
  //The SS starts transmitting a periodic DCI from (10*2microsec+1+TDL) slots after step 3 to schedule PUSCH.
  //Check: Does the UE transmit an RRCSetupComplete message within (10*2microsec+1+TDL+K2) slots15 (FDD)/18 (TDD) subframes after successful completion of step 3?
  v_ReceivedNR_SRB_COMMON_IND := f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
															cas_NR_SRB0_RrcPdu_REQ(nr_Cell1,cs_TimingInfo_Now,f_NR_38508_RRCSetup(nr_Cell1, tsc_NR_RRC_TI_Def)),
															car_NR_SRB1_RrcNasPdu_IND(nr_Cell1, cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *), cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected)),
															10, "Step 4" );

    //@siclog "Step 5 - 6" siclog@
  //The SS transmits a SecurityModeCommand message.
  //The SS starts transmitting a periodic DCI from (5*2microsec+1+TDL) slots after step 5 to schedule PUSCH.
  //Check: Does the UE transmit a SecurityModeComplete message within (5*2microsec+1+TDL+K2) 10 (FDD)/13 (TDD) subframes slots after successful completion of step 5?
  v_SecurityParams := f_NR_Security_Get();
  v_SecurityParams := f_NR_Authentication_InitAS (v_SecurityParams, v_ReceivedNR_SRB_COMMON_IND.Signalling.Nas[0].SecurityProtection.NasCount);
  v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(nr_Cell1);
  f_NR_SS_RRC_EnableIntProt_CiphULandDL(nr_Cell1,
										v_SecurityParams.AS_Integrity,
										v_SecurityParams.AS_Ciphering);

  f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
							 cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,
													-,
													cs_38508_RRCSecurityModeCommand(tsc_NR_RRC_TI_Def,
																					cs_38508_SecurityAlgorithmConfig(v_SecurityParams.AS_Ciphering.Algorithm,
																													 v_SecurityParams.AS_Integrity.Algorithm))),
							 car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCSecurityModeComplete(tsc_NR_RRC_TI_Def)),
							 5, "Step 6" );

  f_NR_Security_Set(v_SecurityParams);

    //@siclog "Step 7 - 8" siclog@
  //The SS transmits a UECapabilityEnquiry message.
  //The SS starts transmitting a periodic DCI from (80*2microsec+1+TDL) slots after step 7 to schedule PUSCH.
  //Check: Does the UE transmit a UECapabilityInformation message within (80*2microsec+1+TDL+K2) slots after successful completion of step 7?
  v_EncodeCapRequestNR := bit2oct(encvalue( cs_38508_UECapabilityRequestFilterNR({ cs_NR_BandInformationNR(f_NR_CellInfo_GetBandIndicator(nr_Cell1)) }) ));
  f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
							 cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_UECapabilityEnquiry(tsc_NR_RRC_TI_Def, nr, v_EncodeCapRequestNR)),
							 car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_UECapabilityInformationAny(tsc_NR_RRC_TI_Def)),
							 80, "Step 8" , true);
							 
    //@siclog "Step 9 - 10" siclog@
  //The SS transmits an RRCReconfiguration message to establish SRB2 and DRB #n for the UE
  fl_NR_BuildRRCReconfigExistingDRBs (v_RadioBearerConfig, v_CellGroupConfig, v_SS_Config, nr_Cell1);
  
  v_SecurityParams := f_NR_Security_Get();
  v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(v_CellGroupConfig)));

  //Reconfigure DRB in SS - existing DRBs are not reconfigured during release
  f_NR_SS_CommonRadioBearerConfig(nr_Cell1, v_SS_Config, cs_TimingInfo_Now);
  
  // Configure PDCP security
  v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(nr_Cell1, false);
  f_NR_SS_RRC_EnableIntProt_CiphULandDL(nr_Cell1,
										v_SecurityParams.AS_Integrity,
										v_SecurityParams.AS_Ciphering);

  v_ServerAcceptMsg := {cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, f_Get_NG_ServiceAcceptMsg(v_ReceivedNR_SRB_COMMON_IND.Signalling.Nas[0].Pdu.Msg.service_Request.pduSessionStatus))}; 
  v_RrcMsg := cs_NR_RRC_MSG_Request_DCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig, -, -, v_V1530Ext));
  f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
							 cas_NR_RrcNasPduList_REQ(nr_Cell1, tsc_NR_RbId_SRB1, cs_TimingInfo_Now, v_RrcMsg, v_ServerAcceptMsg),
							 car_NR_SRB_RrcPdu_IND(nr_Cell1, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete),
							 10, "Step 10" );

    //@siclog "Step 11 - 12" siclog@
  //The SS transmits a CounterCheck message.
  //The SS starts transmitting a periodic DCI from (5*2microsec+1+TDL) slots after step 11 to schedule PUSCH.
  //Check: Does the UE transmit a CounterCheckResponse message within (5*2microsec+1+TDL+K2)10 (FDD)/13 (TDD) subframes slots after successful completion of step 11?
  v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();
  v_CountList := {cs_38508_DRB_CountMSB_Info(v_ExistingDRBs[0].DRB_Id, 1, 1)};	  
  v_CountList_UE := {cr_DRB_CountInfo(v_ExistingDRBs[0].DRB_Id)};
  f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
							 cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_CounterCheck(tsc_NR_RRC_TI_Def, v_CountList)),
							 car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_CounterCheckResponse(tsc_NR_RRC_TI_Def, v_CountList_UE))
							 5, "Step 12" );

    //@siclog "Step 13" siclog@
  //Void

    //@siclog "Step 14a1-14a7" siclog@
  if ( pc_inactiveState ){

    //@siclog "Step 14a1" siclog@
  //The SS transmits an RRCRelease message with suspendConfig.
  f_NR_RRC_InactiveDef(nr_Cell1);

    //@siclog "Step 14a2" siclog@
  //Wait 1s to let UE enter RRC_INACTIVE state.
  f_Delay(1.0);

    //@siclog "Step 14a3" siclog@
  //The SS transmits a Paging message including a matched ng-5G-S-TMSI.
  f_NR_Paging_Def(nr_Cell1, RRC_INACTIVE);

  v_CellGroupConfig := f_NR_GetSCGConfigResume(nr_Cell1);
  v_EncodedSCG_Config := bit2oct(encvalue(v_CellGroupConfig));

    //@siclog "Step 14a4" siclog@
  //The UE transmits an RRCResumeRequest message
  SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, mt_Access)));

    //@siclog "Step 14a5" siclog@
  //The SS transmits an RRCResume message for the UE

  if(pc_reducedCP_Latency){
        //@siclog "Step 14a5Aa1 - 14a5Aa2" siclog@
	  //The SS starts transmitting a periodic DCI from (10* 2 microsec +1+TDL) slots after step 14a5 to schedule PUSCH.
	  //Check: Does the UE transmit an RRCResumeComplete message within (6*2microsec+1+TDL+K2) slots after successful completion of step 14a5?
	  f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
								 cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, cs_38508_RRCResume(tsc_NR_RRC_TI_Def, v_EncodedSCG_Config, v_RadioBearerConfig)),
								 car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeComplete),
								 6, "Step 14a5Aa2" );
  }else
  {
        //@siclog "Step 14a5Ab1 - 14a5Ab2" siclog@
	  //The SS starts transmitting a periodic DCI from n slots after step 14a5 to schedule PUSCH.
	  //Check: Does the UE transmit an RRCResumeComplete message within (10* 2 microsed +1+TDL+K2)slots after successful completion of step 14a5?
	  f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
								 cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, cs_38508_RRCResume(tsc_NR_RRC_TI_Def, v_EncodedSCG_Config, v_RadioBearerConfig)),
								 car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeComplete),
								 10, "Step 14a5Ab2" );
  }
  v_NR_RachProcedureConfig := f_NR_InitialiseRachProcedure(nr_Cell1);
  f_NR_SS_RachProcedureConfig(nr_Cell1, v_NR_RachProcedureConfig);

    //@siclog "Step 14a6-14a7" siclog@
  //Void
  }
    //@siclog "Step 14-21" siclog@
  //Void

    //@siclog "Step 22" siclog@
  //The SS re-adjusts the SS levels according to row "T1" in table 8.1.5.8.1.3.2-1/2 in order that the radio link quality of NR Cell 1 is degraded, and NR Cell 2 is suitable for camping
  f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 23" siclog@
  //Wait 1s to let T310 expire
  f_Delay(1.0);

    //@siclog "Step 24" siclog@
  //The UE transmits an RRCReestablishmentRequest message on Cell 2

  v_RNTI_Value := f_NR_CellInfo_GetRNTI(nr_Cell1);
  v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
  v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(nr_Cell2);
	  
  f_NR_CellInfo_SetRNTI(nr_Cell2, tsc_C_RNTI_Value4);
  f_NR_SS_C_RNTI_Config(nr_Cell2, tsc_C_RNTI_Value4, -, tsc_NoCnfReq, tsc_FollowOnFlag);
  f_NR_SS_CommonCellConfig(nr_Cell2, cads_NR_RachProcedure_DcchDtchToCSS_Config_REQ(nr_Cell2, cs_TimingInfo_Now, tsc_CnfReq,
																					cs_NR_RachProcedureConfig_CBRA_Msg4Based(v_NR_PhysicalParameters, cs_NR_RachProcedureMsg4WithNoRrcMsg),
																					cs_NR_SearchSpaceDlDciAssignment_CSSType1(v_NR_PhysicalParameters),
																					cs_NR_SearchSpaceUlDciAssignmentCSS(v_NR_PhysicalParameters, cs_NR_DciFormat_0_0_Params)));

  v_ShortMAC_I := f_NR_Calculate_ShortMAC(nr_Cell1, nr_Cell2);

  SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell2, cr_38508_RRCReestablishmentRequest(cr_NR_ReestabUE_Identity (v_RNTI_Value, v_PhysCellId, v_ShortMAC_I), otherFailure)));
  

    //@siclog "Step 25 - 26 " siclog@
  //The SS transmits an RRCReestablishment message
  //The SS starts transmitting a periodic DCI from (10*2microsec+1+TDL) slots after step 25 to schedule PUSCH.
  //Check: Does the UE transmit an RRCReestablishmentComplete message within (10*2microsec+1+TDL+K2) 15 (FDD)/18 (TDD) subframes slots after successful completion of step 26?
 
  f_NR_ULGrantConfiguration_Start(nr_Cell2, -, -, -, cs_NR_UplinkTimeAlignment_Stop);
  f_NR_SS_AS_ActivateSecurity_Current(nr_Cell2, 0, -, true);
  
f_NR5GC_RRC_ProcedureDelay(nr_Cell2,
							cas_NR_SRB1_RrcPdu_REQ(nr_Cell2, cs_TimingInfo_Now, cs_38508_RRCReestablishment_Def(tsc_NR_RRC_TI_Def, 0)),
							car_NR_SRB1_RrcPdu_IND ( nr_Cell2, cr_38508_RRCReestablishmentComplete),
							10, "Step 26" );

 f_NR5GC_RRCReestablishment_Steps8To13(nr_Cell1, nr_Cell2);
 
   //@siclog "Step 27 - 35" siclog@
 //Void

}


1. [bookmark: _Toc84670987][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 2
	Function name
	f_NR5GC_RRC_ProcedureDelay

	Reason for change
	1. It is wrong toassign a new value to p_SRB_COMMON_REQ, which will not be valid.
2. when SS is dealling with UL message, the watch dog timer maybe too short tol fail the test case.
3. Add for change1 point1.

	Summary of change
	1. Add a new variable v_SRB_COMMON_REQ default value is equal to p_SRB_COMMON_REQ.
2. Add  more 200ms for watch dog timer. 
3. Add a new input parameter p_If_UEinfo.

	TTCN module
	NR5GC_RRCSteps.ttcn

	MCC160 Comment
	


Before Change
	  function f_NR5GC_RRC_ProcedureDelay(NR_CellId_Type                        p_NR_CellId,
                                      template (value) NR_SRB_COMMON_REQ    p_SRB_COMMON_REQ,
                                      template (present) NR_SRB_COMMON_IND  p_SRB_COMMON_IND,
                                      integer                               p_Nms,
                                      charstring                            p_TestcaseStep
                                      ) runs on NR_BASE_PTC return NR_SRB_COMMON_IND
  {
    var integer v_T1Slots;
    var integer v_T2Slots;
    var integer v_T3Slots;
    var integer v_T4Slots;
    var SubFrameTiming_Type v_T1; // Send DL Msg
    var SubFrameTiming_Type v_T2; // HARQ ACK receive
    var SubFrameTiming_Type v_T3; // Start UL Grant start
    var SubFrameTiming_Type v_T4; // UL RRC PDU received
    var SubFrameTiming_Type v_Tnow;
    var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
    var integer v_Nslots := p_Nms*v_SlotsNumberPerSubframe + 1;  //@sic R5s210842 change 2.1 sic@
    var integer v_Delta1;
    var integer v_Delta2;
    var integer v_GrantPeriodicity;
    var integer v_DL_RetransmissionCnt := 0;
    var NR_SYSTEM_IND v_SYS_IND;
    var boolean v_T2Flag := false;
    var boolean v_T4Flag := false;
    var integer v_K1;
    var integer v_K2 := 4;
    var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
    var integer v_MaxNumberSlots := 1024 * 10 * v_SlotsNumberPerSubframe;     // Maximum number per SFN complete cycle

    timer t_Watchdog;
    
    //@sic R5s210842 sic@
    if (f_TimingInfo_IsNow(p_SRB_COMMON_REQ.Common.TimingInfo)) {
        v_T1 := f_NR_GetNextSendOccasion_DL(p_NR_CellId);
        p_SRB_COMMON_REQ.Common.TimingInfo.SubFrame := v_T1;
    } else {
        v_T1 := valueof(p_SRB_COMMON_REQ.Common.TimingInfo.SubFrame);
    }
    v_T1Slots := f_NR_ConvertTiming_To_TimingInSlots(v_T1, v_SlotsNumberPerSubframe);
    // Initialise GrantPeriodicity, Delta1, K1
    if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
        v_GrantPeriodicity := 1; // Grant repeated in every slot
        v_Delta1 := 0;
        v_K1 := 7;
    } else { // TDD
        select(v_SCS) {
            case(kHz15) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5 - (v_T1Slots + v_Nslots)) mod 5;
                v_K1 := 7;
            }
            case(kHz30) {
                v_GrantPeriodicity := 10; //Grant repeated every 10 slots
                v_Delta1 := (4 - (v_T1Slots + v_Nslots) mod 10) mod 10;
                v_K1 := 5;
            }
            case(kHz120) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5- (v_T1Slots + v_Nslots) mod 5) mod 5;
                v_K1 := 5;
            }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    }
    
    //Calculate T3
    v_T3Slots := v_T1Slots + v_Nslots + v_Delta1;  //@sic R5s210842 change 2 sic@
    v_T3 := f_NR_ConvertTimingInSlots_To_Timing(v_T3Slots, v_SCS, v_SlotsNumberPerSubframe);
    
    //T1: send DL Msg
    SRB.send(p_SRB_COMMON_REQ);

    //Enable HARQ ack/nack reporting
    f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(enable), cs_TimingInfo_NR(v_T1)); //@sic R5s210842 change 2.4 sic@
    
    //T3 configure SS with periodic UL Grant at T3
    f_NR_ULGrantConfiguration_Start(p_NR_CellId, cs_TimingInfo_NR(v_T3), -, -, -, cs_UL_GrantConfig_ContPeriodic(v_GrantPeriodicity));
                                          
    // Start WatchDog timer
    v_Tnow := f_NR_GetCurrentTiming(p_NR_CellId);
    t_Watchdog.start (int2float(f_SubFrameTiming_Duration(v_Tnow, v_T1) + p_Nms + 50) / 1000.0);  /* 50ms added to be at the safe side */
    
    alt {
        /* HARQ NACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, nack)) {
            v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;
            if (v_DL_RetransmissionCnt > 4) {
                t_Watchdog.stop;
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "DL Msg retransmissions > 4");
            } else {
                repeat;
            }
        }
        /* HARQ ACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, ack))-> value v_SYS_IND {
            v_T2 := v_SYS_IND.Common.TimingInfo.SubFrame;
            f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(disable));    /* disable indication of HARQ ack/nack; */
            v_T2Flag := true;
            repeat;
        }
        /* HARQ error indication */
        [] SYSIND.receive (car_NR_SYSTEM_HARQ_ERROR(p_NR_CellId, cr_NR_HarqErrorUL_Any)) {
            f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in RRC Delay test");
            t_Watchdog.stop;
        }
        [] SRB.receive(p_SRB_COMMON_IND) -> value v_ReceivedNR_SRB_COMMON_IND {
            v_T4 := v_ReceivedNR_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
            v_T4Flag := true;
            t_Watchdog.stop;
        }
        [] t_Watchdog.timeout {                                                                                     /* timeout => after T5 */
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "t_Watchdog timeout");
        }
    }
    //Stop UL Grant
    f_NR_ULGrantConfiguration_Stop(p_NR_CellId);
    
    if (not v_T2Flag) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL ACK not received");
    }
    if (not v_T4Flag) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL Msg not received");
    }
    
    //Check the Delay
    v_T2Slots := (((v_T2.SFN.Number * 10) + v_T2.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T2) + v_MaxNumberSlots;
    v_T4Slots := (((v_T4.SFN.Number * 10) + v_T4.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T4) + v_MaxNumberSlots;
    //Initialise Delta2
    if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
        v_Delta2 := 0;
    } else { //TDD
        select(v_SCS) {
            case(kHz15) {
                v_Delta2 := (5 - (v_T2Slots - v_K1 + v_Nslots)) mod 5;
            }
            case(kHz30) {
                v_Delta2 := (4 - (v_T2Slots - v_K1 + v_Nslots) mod 10) mod 10;
            }
            case(kHz120) {
                v_Delta2 := (5- (v_T2Slots - v_K1 + v_Nslots) mod 5) mod 5;
            }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    }
    if ( ((v_T4Slots - v_K2) - (v_T2Slots - v_K1))mod v_MaxNumberSlots <= (v_Nslots + v_Delta2)) {
        f_NR_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);
    } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);
    }

   return v_ReceivedNR_SRB_COMMON_IND;
  }


After Change
	function f_NR5GC_RRC_ProcedureDelay(NR_CellId_Type						p_NR_CellId,
									  template (value) NR_SRB_COMMON_REQ	p_SRB_COMMON_REQ,
									  template (present) NR_SRB_COMMON_IND	p_SRB_COMMON_IND,
									  integer								p_Nms,
   	  								  charstring							p_TestcaseStep,
   	  								  boolean p_If_UEinfo := false) runs on NR_BASE_PTC return NR_SRB_COMMON_IND 
  {
	var integer v_T1Slots;
	var integer v_T2Slots;
	var integer v_T3Slots;
	var integer v_T4Slots;
	var SubFrameTiming_Type v_T1; // Send DL Msg
	var SubFrameTiming_Type v_T2; // HARQ ACK receive
	var SubFrameTiming_Type v_T3; // Start UL Grant start
	var SubFrameTiming_Type v_T4; // UL RRC PDU received
	var SubFrameTiming_Type v_Tnow;
	var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
	var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
	var integer v_Nslots := p_Nms*v_SlotsNumberPerSubframe + 1;  //@sic R5s210842 change 2.1 sic@
	var integer v_Delta1;
	var integer v_Delta2;
	var integer v_GrantPeriodicity;
	var integer v_DL_RetransmissionCnt := 0;
	var NR_SYSTEM_IND v_SYS_IND;
	var boolean v_T2Flag := false;
	var boolean v_T4Flag := false;
	var integer v_K1;
	var integer v_K2 := 4;
	var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
	var integer v_MaxNumberSlots := 1024 * 10 * v_SlotsNumberPerSubframe;	  // Maximum number per SFN complete cycle
	var template (value) NR_SRB_COMMON_REQ v_SRB_COMMON_REQ := p_SRB_COMMON_REQ;
	timer t_Watchdog;
		
	//@sic R5s210842 sic@
	if (f_TimingInfo_IsNow(v_SRB_COMMON_REQ.Common.TimingInfo)) {
		v_T1 := f_NR_GetNextSendOccasion_DL(p_NR_CellId);
		v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame := v_T1;
	} else {
		v_T1 := valueof(v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame);
	}
	v_T1Slots := f_NR_ConvertTiming_To_TimingInSlots(v_T1, v_SlotsNumberPerSubframe);
	// Initialise GrantPeriodicity, Delta1, K1
	if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
		v_GrantPeriodicity := 1; // Grant repeated in every slot
		v_Delta1 := 0;
		v_K1 := 7;
	} else { // TDD
		select(v_SCS) {
			case(kHz15) {
				v_GrantPeriodicity := 5; //Grant repeated every 5 slots
				v_Delta1 := (5 - (v_T1Slots + v_Nslots)) mod 5;
				v_K1 := 7;
			}
			case(kHz30) {
				v_GrantPeriodicity := 10; //Grant repeated every 10 slots
				v_Delta1 := (4 - (v_T1Slots + v_Nslots) mod 10) mod 10;
				v_K1 := 5;
			}
			case(kHz120) {
				v_GrantPeriodicity := 5; //Grant repeated every 5 slots
				v_Delta1 := (5- (v_T1Slots + v_Nslots) mod 5) mod 5;
				v_K1 := 5;
			}
			case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
		}
	}
	
	//Calculate T3
	v_T3Slots := v_T1Slots + v_Nslots + v_Delta1;  //@sic R5s210842 change 2 sic@
	v_T3 := f_NR_ConvertTimingInSlots_To_Timing(v_T3Slots, v_SCS, v_SlotsNumberPerSubframe);
	
	//T1: send DL Msg
	SRB.send(v_SRB_COMMON_REQ);
  
	//Enable HARQ ack/nack reporting
	f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(enable), cs_TimingInfo_NR(v_T1)); //@sic R5s210842 change 2.4 sic@
	
	//T3 configure SS with periodic UL Grant at T3
	f_NR_ULGrantConfiguration_Start(p_NR_CellId, cs_TimingInfo_NR(v_T3), -, -, -, cs_UL_GrantConfig_ContPeriodic(v_GrantPeriodicity));
										  
	// Start WatchDog timer
	v_Tnow := f_NR_GetCurrentTiming(p_NR_CellId);
  t_Watchdog.start (int2float(f_SubFrameTiming_Duration(v_Tnow, v_T1) + p_Nms + 250) / 1000.0);  /* 50ms added to be at the safe side */ 
	
	alt {
		/* HARQ NACK */
		[]SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, nack)) {
			v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;
			if (v_DL_RetransmissionCnt > 4) {
				t_Watchdog.stop;
				f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "DL Msg retransmissions > 4");
			} else {
				repeat;
			}
		}
		/* HARQ ACK */
		[]SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, ack))-> value v_SYS_IND {
			v_T2 := v_SYS_IND.Common.TimingInfo.SubFrame;
			f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(disable));	/* disable indication of HARQ ack/nack; */
			v_T2Flag := true;
			repeat;
		}
		/* HARQ error indication */
		[] SYSIND.receive (car_NR_SYSTEM_HARQ_ERROR(p_NR_CellId, cr_NR_HarqErrorUL_Any)) {
			f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in RRC Delay test");
			t_Watchdog.stop;
		}
		[] SRB.receive(p_SRB_COMMON_IND) -> value v_ReceivedNR_SRB_COMMON_IND {
			v_T4 := v_ReceivedNR_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
			v_T4Flag := true;
			t_Watchdog.stop;
		}
		[] t_Watchdog.timeout { 																					/* timeout => after T5 */
			f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "t_Watchdog timeout");
		}
	}
	//Stop UL Grant
	f_NR_ULGrantConfiguration_Stop(p_NR_CellId);
	
	if (not v_T2Flag) {
		f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL ACK not received");
	}
	if (not v_T4Flag) {
		f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL Msg not received");
	}
	
	//Check the Delay
	v_T2Slots := (((v_T2.SFN.Number * 10) + v_T2.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T2) + v_MaxNumberSlots;
	v_T4Slots := (((v_T4.SFN.Number * 10) + v_T4.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T4) + v_MaxNumberSlots;
	//Initialise Delta2
	if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
		v_Delta2 := 0;
	} else { //TDD
		select(v_SCS) {
			case(kHz15) {
				v_Delta2 := (5 - (v_T2Slots - v_K1 + v_Nslots)) mod 5;
			}
			case(kHz30) {
				v_Delta2 := (4 - (v_T2Slots - v_K1 + v_Nslots) mod 10) mod 10;
			}
			case(kHz120) {
				v_Delta2 := (5- (v_T2Slots - v_K1 + v_Nslots) mod 5) mod 5;
			}
			case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
		}
	}
	
	if(p_If_UEinfo){//Script failure lc add 20211006 for UE infor is too big 
		v_Nslots := v_Nslots + 20;
	}
	
	if ( ((v_T4Slots - v_K2) - (v_T2Slots - v_K1))mod v_MaxNumberSlots <= (v_Nslots + v_Delta2)) {
		f_NR_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);
	} else {
		f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);
	}
  
   return v_ReceivedNR_SRB_COMMON_IND;
  }


1. [bookmark: _Toc84670988]Branches executed
NR Band 78, nea2, Nia2.
1. [bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc84670989]Execution Log Files
3. [bookmark: _Toc84670990]Media Tek M80
The Media Tek M80 passed this test case on Datang LinkTester 5G Protocol Conformance Test System ECT9610. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test Case Execution log file:
\ TC_8_1_5_8_1_NR5GC_N78.log
· [bookmark: _GoBack]PIXCIT settings used:
\ Pics(MTK).xml
\ Pixit(TC_8_1_5_8_1_NR5GC).xml

